

## M5. Differential Equations with Maple

### 1. One variable, first order

Here we consider a differential equation of the general form

$$\frac{dx}{dt} = f(t, x)$$

and the corresponding Initial State Problem (ISP)

$$\frac{dx}{dt} = f(t, x) \quad , \quad x(t_0) = x_0 .$$

It is assumed that you know the function  $f$  (as a function of two variables) and the numbers  $t_0$  and  $x_0$ . You wish to find  $x$  as a function of  $t$  so that the equations above both hold.

#### 1.1 The dsolve command

The dsolve command tries to find an analytic (exact) solution.

For example, to solve the differential equation

$$\frac{dx}{dt} = x + \cos t,$$

proceed as follows.

```

> de1 := diff(x(t), t) = x(t) + cos(t);
              de1 :=  $\frac{\partial}{\partial t} x(t) = x(t) + \cos t$ 
> dsolve(de1, x(t));
              x(t) =  $-\frac{1}{2} \cos(t) + \frac{1}{2} \sin(t) + e^t - C1$ 
```

**Notes (i)** I gave the equation a name (*de1*) before invoking the dsolve command. As usual, this is not necessary, but a good idea.

**(ii)** Wherever you mention the dependent variable (in this case  $x$ ), you need to call it  $x(t)$  — to tell Maple that it is in fact a function of  $t$ .

**(iii)**  $\text{diff}(x(t), t)$  is Maple-speak for  $\frac{dx}{dt}$ .

**(iv)** We use an ordinary equal sign ( $=$ ) in the equation itself, and the colon-equal sign ( $:=$ ) to give it a name for later reference.

**(v)** The solution of a differential equation usually has one or more arbitrary constants. Maple uses notation like  $\_C1$ ,  $\_C2$  and so on for these. The underline symbol is part of the name of the constant, not a minus sign. So the solution which Maple comes up with here, translated into more normal notation, is

$$x = -\frac{1}{2} \cos t + \frac{1}{2} \sin t + Ce^t .$$

(I have changed the name of the arbitrary constant to the more normal  $C$ .)

**(vi)** In the `dsolve` command, you have to specify the equation you want solved, then the variable you want solved for. In a one-variable differential equation such as we are discussing at present, this seems unnecessary. The command however can be used to solve systems of several differential equations in several unknowns, and in this more general case, it is necessary to specify which unknowns you want solved for.

Maple may not be able to find an analytic solution. For example

```
> de2 := diff(x(t), t) = x^3+t*x-1;
      de2 :=  $\frac{\partial}{\partial t}x(t) = x^3 + tx - 1$ 
> dsolve(de2, x(t));
>
```

Maple is unable to solve this equation. There are two possible reasons for this

- (1) There is no analytic solution of the equation — that is, a solution, if it exists, cannot be expressed as a formula in terms of standard functions.
  - (2) There is an analytic solution but Maple is not smart enough to find it.
- For the kind of differential equations we are likely to come across in this course, (2) is not very likely.


## 1.2 Working with solutions

Let us suppose we have just solved our first differential equation above, and we want to do some work with the solution — perhaps substitute some  $t$  values in it, or draw a graph. You might think that you could refer to the solution by name, but no!

```
> de1 := diff(x(t), t) = x(t)+cos(t);
      de1 :=  $\frac{\partial}{\partial t}x(t) = x(t) + \cos t$ 
> dsolve(de1, x(t));
      x(t) =  $-\frac{1}{2}\cos(t) + \frac{1}{2}\sin(t) + e^t\_C1$ 
> plot(x(t), t=0..10);
Plotting error, empty plot
> subs(t=0, x);
      x
```

One way round this is to simply type the solution in yourself, giving it a name as you go. One advantage of this is that you can clean up Maple's somewhat peculiar notation as you do.

```
> X := -(1/2)*cos(t)+(1/2)*sin(t)+C*exp(t);
      X :=  $-\frac{1}{2}\cos(t) + \frac{1}{2}\sin(t) + Ce^t$ 
> C := 1;
      C := 1
> plot(X, t=0..10);
```



```
> subs(t=0, X);
      -1/2 cos(0) + 1/2 sin(0) + Ce^0
> simplify(");
      1/2
```

Another approach is to give the solution a name in the `dsolve` command.

```
> sol := dsolve(del , x(t));
      sol := x(t) = -1/2 cos(t) + 1/2 sin(t) + e^t _C1
```

But you will notice that now Maple returns `sol` as the name of the entire equation, not just the function. We could substitute into it if we wished, but we couldn't plot it.

```
> _C1:=1;
      _C1 := 1
> sol;
      sol := x(t) = -1/2 cos(t) + 1/2 sin(t) + e^t
> subs(t=0, sol);
      x(0) = -1/2 cos(0) + 1/2 sin(0) + e^0
> simplify(");
      x(0) = 1/2
```

If you really want to get at the expression for the solution itself (for instance, to plot it), you can use the `op` command. Many of the things that Maple deals with (such a solutions of equations), it stores internally as lists. The `op` command will break the list up into its component parts. So if you ever want a part of something that Maple has given you, it is worth experimenting with `op`. For example, try it on `sol`, the solution found above.

```
> op(1,sol);
      x(t)
> op(2,sol);
      -1/2 cos(t) + 1/2 sin(t) + e^t _C1
```

Aha! All we need to do is to give this a name, and then we can work with it.


```
> X := op(2,sol);
```

$$X := -\frac{1}{2}\cos(t) + \frac{1}{2}\sin(t) + e^t \_C1$$

```
> _C1:=1;
```

$$\_C1 := 1$$

```
> plot(X, t=0..10);
```



And note also,

```
> op(1,X);
```

$$-\frac{1}{2}\cos(t)$$

```
> op(2,X);
```

$$\frac{1}{2}\sin(t)$$

and so on.

### 1.3 The DEplot command

If you cannot get an analytic solution, or even if you can, you can use the DEplot command to get a graph of the solution of the differential equation. More precisely, DEplot draws a graph of a solution of an *ISP* — you need to give it an initial condition to work with.


The DEplot command is part of the DETools package. This means that, before you can use it, you must load the package by telling Maple

```
> with(DEtools);
```

For example, suppose we want to find the solution of the ISP

$$\frac{dx}{dt} = \sin t - \frac{x^2}{5} \quad \text{where} \quad x(0) = 1$$

We could plot it as follows.




The arguments in the DEplot command are

- de3        The actual differential equation to be solved.
- x(t)      The (dependent) variable to be plotted.
- t=0..5    The range of the (independent) variable to plot over.
- {[0,1]}    The initial condition,  $x(0)=1$ .

Note that the initial condition must be surrounded by two types of brackets. This is necessary because you can ask Maple to plot several solutions of the same differential equation, corresponding to different initial conditions, all on the same diagram.

For example




The small red arrows show the direction field of the equation. These give a general idea of the behaviour of any other solutions in the neighbourhood of the ones plotted.

Maple will divide the horizontal range up into twenty steps for its calculations unless you tell it otherwise. This is sometimes not enough for an accurate solution. For example, suppose we want to solve the ISP

$$\frac{dx}{dt} = t \cos x - x \sin t \quad \text{where} \quad x(0) = 2$$


```
> de4 := diff(x(t),t) = t*cos(x(t))-x(t)*sin(t);
      de4 :=  $\frac{\partial}{\partial t}x(t) = t \cos(x(t)) - x(t) \sin(t)$ 
> DEplot(de4, x(t), t=0..10, {[0,2]} );
```



You can see that the graph has been plotted at intervals of 0.4, resulting in sharp “corners” — clearly not very accurate. Also, and much worse, you can see that towards the right hand end of the graph it starts going in a different direction from the direction field arrows (which is impossible): the solution method is losing accuracy badly and the graph is going haywire.


There are several ways of fixing this up. The most obvious one is to ask Maple to use a smaller stepsize. For example

```
> DEplot(de4, x(t), t=0..10, {[0,2]}, stepsize=0.1 );
```



That is obviously much better.

Another way of getting a more accurate plot is to use a more accurate method. Maple can apply a number of different numerical methods for solving differential equations. You can get a more accurate one by using the `method` option.



What does the mysterious option `method=dverk78` actually mean? If you do not tell it otherwise, Maple used what is probably the best-known of the accurate methods of solving differential equations, the Fourth-order Runge-Kutta method. Specifying `method=dverk78` calls for a Seventh-eighth Order Continuous Runge-Kutta method, which is much more accurate. Why not use the more accurate method always? Because it is slower, for one thing: try it and you will see.

A proper discussion of the Runge-Kutta methods (and others) for numerically solving differential equations takes us beyond the scope of the present course.

You can control the appearance of the direction field arrows in several ways. You can specify their shape by `arrows=SMALL`, `arrows=MEDIUM`, `arrows=LARGE`, `arrows=LINE` or `arrows=NONE`. (The `arrows=LINE` option gives you arrows with no heads.) If you do not specify, you get `arrows=SMALL`. You can specify their colour by


`color=green`

and so on, for other colours. If you do not specify, you get `color=red`.

You can specify how many arrows are plotted, in the horizontal and vertical directions separately, by

`dirgrid=[15,10]`

or whatever numbers you want. If you do not specify, you get `dirgrid=[20,20]`.



(The colour effect will not show on printed notes, of course.)

You can specify the colour of the curves themselves by


```
linecolor=black
```

or, if you have several curves and you want to give them different colours, by

```
linecolor=[black,red,blue] ,
```

giving colours in the same order as you gave the initial conditions. (I can't show the colours on this page of course.)

```
> DEplot(de3, x(t), t=0..5 , {[0,0],[0,1],[0,2]},
>     arrows=LINE, color=green, dirgrid=[10,10],
>     linecolor=[black,red,blue]);
```



## 2. Several variable, first order

Now let us consider how to deal with a pair differential equations of the general form

$$\frac{dx}{dt} = f(t, x, y)$$

$$\frac{dy}{dt} = g(t, x, y)$$

and the corresponding Initial State Problem (ISP)

$$\frac{dx}{dt} = f(t, x, y) \quad , \quad x(t_0) = x_0 \quad ,$$

$$\frac{dy}{dt} = g(t, x, y) \quad , \quad y(t_0) = y_0 \quad .$$

It is assumed that you know the functions  $f$  and  $g$  (as functions of three variables) and the numbers  $t_0$  ,  $x_0$  and  $y_0$ . You wish to find  $x$  and  $y$  as functions of  $t$  so that the equations above both hold.

We will also consider a set of three differential equations in three unknowns, of the general form

$$\frac{dx}{dt} = f(t, x, y, z)$$

$$\frac{dy}{dt} = g(t, x, y, z)$$

$$\frac{dz}{dt} = h(t, x, y, z)$$

and the corresponding ISP

$$\frac{dx}{dt} = f(t, x, y, z) \quad , \quad x(t_0) = x_0$$

$$\frac{dy}{dt} = g(t, x, y, z) \quad , \quad y(t_0) = y_0$$

$$\frac{dz}{dt} = h(t, x, y, z) \quad , \quad z(t_0) = z_0 \quad .$$

In fact the methods discussed here can be applied to systems of any number of equations and unknowns, in the obvious way.



## 2.1 The dsolve command

The `dsolve` command can be used to find an analytic solution when one exists. Two problems which can arise are:

- (1) Many sets of DEs simply do not have a nice solution in closed form to be found. In this case `dsolve` gives up.
- (2) Where a solution exists, it might be very complicated. Moreover, Maple often displays it in a more complicated form than necessary.

For example, consider this equations.

$$\frac{dx}{dt} = x + y \quad , \quad \frac{dy}{dt} = x - y.$$

We ask Maple to solve them as follows:

```

> eq1 := [diff(x(t),t)=x(t)+y(t) ,
          diff(y(t),t)=x(t)-y(t) ];
          eq1 := [ \frac{\partial}{\partial t}x(t) = x(t)+y(t) , \frac{\partial}{\partial t}y(t) = x(t)-y(t) ]
> dsolve(eq1, [x(t),y(t)]);
{x(t) = \frac{1}{2} - C1e^{(\sqrt{2}t)} - \frac{1}{4} - C1\sqrt{2}e^{(-\sqrt{2}t)} + \frac{1}{4} - C1\sqrt{2}e^{(\sqrt{2}t)} + \frac{1}{2} - C1e^{(-\sqrt{2}t)}
- \frac{1}{4} - C2\sqrt{2}e^{(-\sqrt{2}t)} + \frac{1}{4} - C2\sqrt{2}e^{(\sqrt{2}t)} , y(t) = -\frac{1}{4} - C1\sqrt{2}e^{(-\sqrt{2}t)} + \frac{1}{4} - C1\sqrt{2}e^{(\sqrt{2}t)} + \frac{1}{2} - C2e^{(\sqrt{2}t)}
+ \frac{1}{4} - C2\sqrt{2}e^{(-\sqrt{2}t)} - \frac{1}{4} - C2\sqrt{2}e^{(\sqrt{2}t)} + \frac{1}{2} - C2e^{(-\sqrt{2}t)} }

```

We have a solution, but it does not look very nice. However, you will notice that it simplifies considerably. For one thing, it has those nasty looking Maple arbitrary constants, `_C1` and `_C2`. If you replace them by `A` and `B` say and then do some collecting, you get a much neater looking form.

$$x(t) = A\left(\frac{2+\sqrt{2}}{4}e^{\sqrt{2}t} + \frac{2-\sqrt{2}}{4}e^{-\sqrt{2}t}\right) + \frac{\sqrt{2}}{4}B(e^{\sqrt{2}t} - e^{-\sqrt{2}t})$$

$$y(t) = \frac{\sqrt{2}}{4}A(e^{\sqrt{2}t} - e^{-\sqrt{2}t}) + B\left(\frac{2-\sqrt{2}}{4}e^{\sqrt{2}t} + \frac{2+\sqrt{2}}{4}e^{-\sqrt{2}t}\right)$$

Remembering that `A` and `B` are arbitrary constants, you can simplify even further.

The point of these last calculations is not to study these particular equations and their solutions in detail, but simply to point out that Maple will often give you a horrible looking jumble as its solution, and it is well worth copying it out and simplifying it. (You can sometimes get Maple itself to help in the simplifying process of course.)

Now try a new example: we make a very small change to the last pair of equations.

```

> eq2 := [diff(x(t),t)=x(t)+y(t) ,
          diff(y(t),t)=x(t)-y(t)^2 ];
          eq2 := [ \frac{\partial}{\partial t}x(t) = x(t)+y(t) , \frac{\partial}{\partial t}y(t) = x(t)-y(t)^2 ]
> dsolve(eq2, [x(t),y(t)]);
>

```

Maple gives up: there is no analytic solution.

## 1.2 The DEplot command


As with single differential equations, if you cannot get an analytic solution with `dsolve`, you can usually get a graph of a solution using `DEplot`. And again, you must choose an initial condition to define a curve (or set of initial conditions to define a set of curves).

Let us start with a simple set of three equations.

$$\frac{dx}{dt} = y, \quad \frac{dy}{dt} = z, \quad \frac{dz}{dt} = -x$$

Since we are going to use `DEplot`, we will need to choose some initial conditions. Let us say that, at time  $t=0$ ,  $x=y=z=1$ .


```
> eq3 := [diff(x(t),t) = y(t) ,  
          diff(y(t),t) = z(t) ,  
          diff(z(t),t) = -x(t) ];  
eq3 := [  $\frac{\partial}{\partial t}x(t) = y(t)$  ,  $\frac{\partial}{\partial t}y(t) = z(t)$  ,  $\frac{\partial}{\partial t}z(t) = -x(t)$  ]  
> DEplot(eq3 , [x,y,z] , t=0..5 , {[0,1,1,1]} , scene=[t,x] );
```




The arguments in the `DEplot` command here are

- `eq3`            The name of the set of equations to be plotted;
- `[x,y,z]`        All the (dependent) variables in the equations;
- `t=0..5`        A range of the independent variable to plot over;
- `{[0,1,1,1]}`    The initial condition:  $t=0$  first, then  $x=1$ ,  $y=1$  and  $z=1$ .
- `scene=[t,x]`    The variable to be plotted: in this case we are asking for a graph of  $x$  against  $t$  (that is,  $t$  on the horizontal and  $x$  on the vertical axis).

You can plot how the other variables vary with time by changing the scene option.



You can also use the scene option to plot the *orbits*, i.e. plot the dependent variables against one another. For example to get a graph with  $x$  on the horizontal and  $y$  on the vertical axis, use `scene=[x,y]`. This gives a parametric graph, with parameter  $t$ .



When there are more than two unknown variables, as in this example, the idea of a direction field does not make sense, so Maple does not attempt to show it.

For some of the applications we will look at, we have a *pair* of equations involving *two* unknowns. Unfortunately, in just this case (i.e. exactly 2 equations), there is a bug in Maple which we have to avoid. It is possible to get around it without too much difficulty, but you need to know that it is there, and what to do about it. The problem is with the `scene` option: you cannot specify a scene option which involves the independent variable  $t$ , so you cannot plot  $x$  against  $t$  or  $y$  against  $t$ . But of course that is just what you want to do much of the time. Let us look first at what happens, then how to get around it.

As an example, consider the ISP

$$\begin{aligned} \frac{dx}{dt} &= y, & x(0) &= 1 \\ \frac{dy}{dt} &= -x, & y(0) &= 1. \end{aligned}$$


(This is one version of the equation(s) of Simple Harmonic Motion, which we discuss in class).

We will ask Maple to plot  $x$  against  $t$ .

```
> eq4 := [diff(x(t),t) = y(t), diff(y(t),t) = -x(t)];
          eq4 := [∂x(t)/∂t = y(t), ∂y(t)/∂t = -x(t)]
> DEplot(eq4, [x(t),y(t)], t=0..5, {[0,1,1]}, scene=[t,x(t)]);
Error, (in DEtools/DEplot/CheckDE) invalid subscript selector
```

This is not very helpful. At least we can take comfort from the fact that it is Maple making the error, not us. The same thing happens if you want a plot of  $y$  against  $t$ . But if we ask for a plot of  $y$  against  $x$ , we get a correct graph!

```
> DEplot(eq4, [x(t),y(t)], t=0..5, {[0,1,1]}, scene=[x(t),y(t)]);
```




What should we do to get graphs of  $x$  or  $y$  against  $t$ . There are two ways of getting around the bug that I know of.

(1) Since the problem only occurs with exactly two equations in two unknowns, add an extra equation (in an extra unknown) that does not interfere with the two you are interested in. You will have to add an extra initial condition for the new variable too. My favourite is to add

$$\frac{dz}{dt} = 1, \quad z(0) = 0.$$

This will make  $z=t$  always, but have no effect on the other equations. Doing this for our pair of equations above,



Note how I have changed the arguments in the DEplot command here to take account of the extra “dummy” variable  $z$ .

(2) For some inexplicable reason, the problem goes away if you add the extra option `arrows=NONE`. If you use this trick you do not have to bother with the third dummy equation.

```
> eq4 := [diff(x(t),t) = y(t) , diff(y(t),t) = -x(t) ];
          eq4 := [ [∂/∂t x(t) = y(t) , ∂/∂t y(t) = -x(t) ]
> DEplot(eq4 , [x(t),y(t)] , t=0..5 , {[0,1,1]} , scene=[t,x(t)] , arrows=NONE );
```

(GIVES SAME DIAGRAM AS ABOVE)

(3) Remember the bug only occurs when the independent variable ( $t$  in these examples) is involved in the scene, so if you only want an orbit (an  $x$ - $y$  plot), you do not need to bother with any of this stuff. In fact, if you do not bother to set the scene, Maple will automatically give you an  $x$ - $y$  plot (with direction field if the equations are autonomous).

```
> DEplot(eq4 , [x(t),y(t)] , t=0..5 , {[0,1,1]} )
```

(GIVES A PLOT OF  $y$  AGAINST  $x$ )

(4) Also, as far as my experiments can tell, the bug only occurs with autonomous systems. If you have a non-autonomous system (i.e. the independent variable  $t$  occurs explicitly on the right hand side of your differential equations), then none of these problems arise. For example, consider the system

$$\begin{aligned} \frac{dx}{dt} &= y & , & & x(0) &= 1 \\ \frac{dy}{dt} &= -tx & , & & y(0) &= 1 . \end{aligned}$$

(This is almost the same as the last system we have been considering, except that I have added a factor  $t$  on the right hand side of the second equation. This change makes the system non-autonomous.)

In this example, the bug does not occur.

```
> eq5 := [diff(x(t),t) = y(t) , diff(y(t),t) = -t*x(t) ];  
eq5 :=  $\left[ \frac{\partial}{\partial t} x(t) = y(t) , \frac{\partial}{\partial t} y(t) = -tx(t) \right]$   
> DEplot(eq5 , [x(t),y(t)] , t=0..5 , {[0,1,1]} , scene=[t,x]);
```

